// File: jynx.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.12;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/utils/math/SafeMath.sol";
import "@uniswap/v2-core/contracts/interfaces/IUniswapV2Factory.sol";
import "@uniswap/v2-core/contracts/interfaces/IUniswapV2Pair.sol";
import "@uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router02.sol";

contract Superstitious is ERC20, Ownable {
   using SafeMath for uint256;

   modifier lockSwap() {
      _inSwap = true;
      _;
      _inSwap = false;
   }

   modifier liquidityAdd() {
      _inLiquidityAdd = true;
      _;
      _inLiquidityAdd = false;
   }

   // == CONSTANTS ==
   uint256 public constant MAX_SUPPLY = 1_000_000_000 ether;
   uint256 public constant BPS_DENOMINATOR = 10_000;
   uint256 public constant SNIPE_BLOCKS = 1;

   // == TAXES ==
   //- @notice Buy devTax in BPS
   uint256 public buyDevTax = 200; //- 2%
   //- @notice Buy rewardsTax in BPS
   uint256 public buyRewardsTax = 0;
   //- @notice Sell devTax in BPS
   uint256 public sellDevTax = 200; //- 2%
   //- @notice Sell rewardsTax in BPS
   uint256 public sellRewardsTax = 0;
   //- @notice address that devTax is sent to
   address payable public devTaxRecipient;
   //- @notice address that rewardsTax is sent to
   address payable public rewardsTaxRecipient;
   //- @notice tokens currently allocated for devTax
   uint256 public totalDevTax;
   //- @notice tokens currently allocated for rewardsTax
   uint256 public totalRewardsTax;

   // == FLAGS ==
   //- @notice flag indicating whether initialDistribute() was successfully called
   bool public initialDistributeDone = false;
   //- @notice flag indicating Uniswap trading status
   bool public tradingActive = false;
   //- @notice flag indicating swapAll enabled
   bool public swapFees = true;

   // == UNISWAP ==
   IUniswapV2Router02 public router;
   address public pair;

   // == WALLET STATUSES ==
   //- @notice Maps each recipient to their tax exlcusion status
   mapping(address => bool) public taxExcluded;
   //- @notice Maps each recipient to the last timestamp they bought
   mapping(address => uint256) public lastBuy;
   //- @notice Maps each recipient to their blacklist status
   mapping(address => bool) public blacklist;

   // == MISC ==
   //- @notice Block when trading is first enabled
   uint256 public tradingBlock;
   //- @notice Contract token balance threshold before `_swap` is invoked
   uint256 public minTokenBalance = 1000 ether;

   // == INTERNAL ==
   uint256 internal _totalSupply = 0;
   bool internal _inSwap = false;
   bool internal _inLiquidityAdd = false;
   mapping(address => uint256) private _balances;

   event DevTaxRecipientChanged(
      address previousRecipient,
      address nextRecipient
   );
   event RewardsTaxRecipientChanged(
      address previousRecipient,
      address nextRecipient
   );
   event BuyDevTaxChanged(uint256 previousTax, uint256 nextTax);
   event SellDevTaxChanged(uint256 previousTax, uint256 nextTax);
   event BuyRewardsTaxChanged(uint256 previousTax, uint256 nextTax);
   event SellRewardsTaxChanged(uint256 previousTax, uint256 nextTax);
   event DevTaxRescued(uint256 amount);
   event RewardsTaxRescued(uint256 amount);
   event TradingActiveChanged(bool enabled);
   event TaxExclusionChanged(address user, bool taxExcluded);
   event BlacklistUpdated(address user, bool previousStatus, bool nextStatus);
   event SwapFeesChanged(bool previousStatus, bool nextStatus);

   constructor(
      address _factory,
      address _router,
      address payable _devTaxRecipient,
      address payable _rewardsTaxRecipient
   ) ERC20("Superstitious", "JYNX") Ownable() {
      taxExcluded[owner()] = true;
      taxExcluded[address(0)] = true;
      taxExcluded[_devTaxRecipient] = true;
      taxExcluded[_rewardsTaxRecipient] = true;
      taxExcluded[address(this)] = true;

      devTaxRecipient = _devTaxRecipient;
      rewardsTaxRecipient = _rewardsTaxRecipient;

      router = IUniswapV2Router02(_router);
      IUniswapV2Factory factory = IUniswapV2Factory(_factory);
      pair = factory.createPair(address(this), router.WETH());

      _mint(msg.sender, MAX_SUPPLY);
   }

   function addLiquidity(uint256 tokens)
      external
      payable
      onlyOwner
      liquidityAdd
   {
      _rawTransfer(msg.sender, address(this), tokens);
      _approve(address(this), address(router), tokens);

      router.addLiquidityETH{value: msg.value}(
         address(this),
         tokens,
         0,
         0,
         owner(),
         // solhint-disable-next-line not-rely-on-time
         block.timestamp
      );
   }

   //- @notice Change the address of the devTax recipient
   //- @param _devTaxRecipient The new address of the devTax recipient
   function setDevTaxRecipient(address payable _devTaxRecipient)
      external
      onlyOwner
   {
      emit DevTaxRecipientChanged(devTaxRecipient, _devTaxRecipient);
      devTaxRecipient = _devTaxRecipient;
   }

   //- @notice Change the address of the rewardTax recipient
   //- @param _rewardsTaxRecipient The new address of the rewardTax recipient
   function setRewardsTaxRecipient(address payable _rewardsTaxRecipient)
      external
      onlyOwner
   {
      emit RewardsTaxRecipientChanged(
         rewardsTaxRecipient,
         _rewardsTaxRecipient
      );
      rewardsTaxRecipient = _rewardsTaxRecipient;
   }

   //- @notice Change the buy devTax rate
   //- @param _buyDevTax The new devTax rate
   function setBuyDevTax(uint256 _buyDevTax) external onlyOwner {
      require(
         _buyDevTax <= BPS_DENOMINATOR,
         "_buyDevTax cannot exceed BPS_DENOMINATOR"
      );
      emit BuyDevTaxChanged(buyDevTax, _buyDevTax);
      buyDevTax = _buyDevTax;
   }

   //- @notice Change the buy devTax rate
   //- @param _sellDevTax The new devTax rate
   function setSellDevTax(uint256 _sellDevTax) external onlyOwner {
      require(
         _sellDevTax <= BPS_DENOMINATOR,
         "_sellDevTax cannot exceed BPS_DENOMINATOR"
      );
      emit SellDevTaxChanged(sellDevTax, _sellDevTax);
      sellDevTax = _sellDevTax;
   }

   //- @notice Change the buy rewardsTax rate
   //- @param _buyRewardsTax The new buy rewardsTax rate
   function setBuyRewardsTax(uint256 _buyRewardsTax) external onlyOwner {
      require(
         _buyRewardsTax <= BPS_DENOMINATOR,
         "_buyRewardsTax cannot exceed BPS_DENOMINATOR"
      );
      emit BuyRewardsTaxChanged(buyRewardsTax, _buyRewardsTax);
      buyRewardsTax = _buyRewardsTax;
   }

   //- @notice Change the sell rewardsTax rate
   //- @param _sellRewardsTax The new sell rewardsTax rate
   function setSellRewardsTax(uint256 _sellRewardsTax) external onlyOwner {
      require(
         _sellRewardsTax <= BPS_DENOMINATOR,
         "_sellRewardsTax cannot exceed BPS_DENOMINATOR"
      );
      emit SellRewardsTaxChanged(sellRewardsTax, _sellRewardsTax);
      sellRewardsTax = _sellRewardsTax;
   }

   //- @notice Rescue ATI from the devTax amount
   //- @dev Should only be used in an emergency
   //- @param _amount The amount of ATI to rescue
   //- @param _recipient The recipient of the rescued ATI
   function rescueDevTaxTokens(uint256 _amount, address _recipient)
      external
      onlyOwner
   {
      require(
         _amount <= totalDevTax,
         "Amount cannot be greater than totalDevTax"
      );
      _rawTransfer(address(this), _recipient, _amount);
      emit DevTaxRescued(_amount);
      totalDevTax -= _amount;
   }

   //- @notice Rescue ATI from the rewardsTax amount
   //- @dev Should only be used in an emergency
   //- @param _amount The amount of ATI to rescue
   //- @param _recipient The recipient of the rescued ATI
   function rescueRewardsTaxTokens(uint256 _amount, address _recipient)
      external
      onlyOwner
   {
      require(
         _amount <= totalRewardsTax,
         "Amount cannot be greater than totalRewardsTax"
      );
      _rawTransfer(address(this), _recipient, _amount);
      emit RewardsTaxRescued(_amount);
      totalRewardsTax -= _amount;
   }

   //- @notice Admin function to update a recipient's blacklist status
   //- @param user the recipient
   //- @param status the new status
   function updateBlacklist(address user, bool status)
      external
      virtual
      onlyOwner
   {
      _updateBlacklist(user, status);
   }

   function _updateBlacklist(address user, bool status) internal virtual {
      emit BlacklistUpdated(user, blacklist[user], status);
      blacklist[user] = status;
   }

   //- @notice Enables trading on Uniswap
   function enableTrading() external onlyOwner {
      tradingActive = true;
   }

   //- @notice Updates tax exclusion status
   //- @param _account Account to update the tax exclusion status of
   //- @param _taxExcluded If true, exclude taxes for this user
   function setTaxExcluded(address _account, bool _taxExcluded)
      public
      onlyOwner
   {
      taxExcluded[_account] = _taxExcluded;
      emit TaxExclusionChanged(_account, _taxExcluded);
   }

   //- @notice Enable or disable whether swap occurs during `_transfer`
   //- @param _swapFees If true, enables swap during `_transfer`
   function setSwapFees(bool _swapFees) external onlyOwner {
      emit SwapFeesChanged(swapFees, _swapFees);
      swapFees = _swapFees;
   }

   function balanceOf(address account)
      public
      view
      virtual
      override
      returns (uint256)
   {
      return _balances[account];
   }

   function _addBalance(address account, uint256 amount) internal {
      _balances[account] = _balances[account] + amount;
   }

   function _subtractBalance(address account, uint256 amount) internal {
      _balances[account] = _balances[account] - amount;
   }

   function _transfer(
      address sender,
      address recipient,
      uint256 amount
   ) internal override {
      require(!blacklist[recipient], "Recipient is blacklisted");

      if (taxExcluded[sender] || taxExcluded[recipient]) {
         _rawTransfer(sender, recipient, amount);
         return;
      }

      bool overMinTokenBalance = balanceOf(address(this)) >= minTokenBalance;
      if (overMinTokenBalance && !_inSwap && sender != pair && swapFees) {
         swapAll();
      }

      uint256 send = amount;
      uint256 devTax;
      uint256 rewardsTax;
      if (sender == pair) {
         require(tradingActive, "Trading is not yet active");
         if (block.number <= tradingBlock + SNIPE_BLOCKS) {
            _updateBlacklist(recipient, true);
         }
         (send, devTax, rewardsTax) = _getTaxAmounts(amount, true);
      } else if (recipient == pair) {
         require(tradingActive, "Trading is not yet active");
         (send, devTax, rewardsTax) = _getTaxAmounts(amount, false);
      } 
      _rawTransfer(sender, recipient, send);
      _takeTaxes(sender, devTax, rewardsTax);
   }

   //- @notice Peforms auto liquidity and tax distribution
   function swapAll() public {
      if (!_inSwap) {
         _swap(balanceOf(address(this)));
      }
   }

   //- @notice Perform a Uniswap v2 swap from token to ETH and handle tax distribution
   //- @param amount The amount of token to swap in wei
   //- @dev `amount` is always <= this contract's ETH balance.
   function _swap(uint256 amount) internal lockSwap {
      address[] memory path = new address[](2);
      path[0] = address(this);
      path[1] = router.WETH();

      _approve(address(this), address(router), amount);

      uint256 contractEthBalance = address(this).balance;

      router.swapExactTokensForETHSupportingFeeOnTransferTokens(
         amount,
         0,
         path,
         address(this),
         block.timestamp
      );

      uint256 tradeValue = address(this).balance - contractEthBalance;

      uint256 totalTaxes = totalDevTax.add(totalRewardsTax);
      uint256 devAmount = amount.mul(totalDevTax).div(totalTaxes);
      uint256 rewardsAmount = amount.mul(totalRewardsTax).div(totalTaxes);

      uint256 devEth = tradeValue.mul(totalDevTax).div(totalTaxes);
      uint256 rewardsEth = tradeValue.mul(totalRewardsTax).div(totalTaxes);

      // Update state
      totalDevTax = totalDevTax.sub(devAmount);
      totalRewardsTax = totalRewardsTax.sub(rewardsAmount);

      // Do transfer
      if (devEth > 0) {
         devTaxRecipient.transfer(devEth);
      }
      if (rewardsEth > 0) {
         rewardsTaxRecipient.transfer(rewardsEth);
      }
   }

   //- @notice Change the minimum contract ACAP balance before `_swap` gets invoked
   //- @param _minTokenBalance The new minimum balance
   function setMinTokenBalance(uint256 _minTokenBalance) external onlyOwner {
      minTokenBalance = _minTokenBalance;
   }

   //- @notice Admin function to rescue ETH from the contract
   function rescueETH() external onlyOwner {
      payable(owner()).transfer(address(this).balance);
   }

   //- @notice Transfers ATI from an account to this contract for taxes
   //- @param _account The account to transfer ATI from
   //- @param _devTaxAmount The amount of devTax tax to transfer
   function _takeTaxes(
      address _account,
      uint256 _devTaxAmount,
      uint256 _rewardsTaxAmount
   ) internal {
      require(_account != address(0), "taxation from the zero address");

      uint256 totalAmount = _devTaxAmount.add(_rewardsTaxAmount);
      _rawTransfer(_account, address(this), totalAmount);
      totalDevTax += _devTaxAmount;
      totalRewardsTax += _rewardsTaxAmount;
   }

   //- @notice Get a breakdown of send and tax amounts
   //- @param amount The amount to tax in wei
   //- @return send The raw amount to send
   //- @return devTax The raw devTax tax amount
   function _getTaxAmounts(uint256 amount, bool buying)
      internal
      view
      returns (
         uint256 send,
         uint256 devTax,
         uint256 rewardsTax
      )
   {
      if (buying) {
         devTax = amount.mul(buyDevTax).div(BPS_DENOMINATOR);
         rewardsTax = amount.mul(buyRewardsTax).div(BPS_DENOMINATOR);
      } else {
         devTax = amount.mul(sellDevTax).div(BPS_DENOMINATOR);
         rewardsTax = amount.mul(sellRewardsTax).div(BPS_DENOMINATOR);
      }
      send = amount.sub(devTax).sub(rewardsTax);
   }

   // modified from OpenZeppelin ERC20
   function _rawTransfer(
      address sender,
      address recipient,
      uint256 amount
   ) internal {
      require(sender != address(0), "transfer from the zero address");
      require(recipient != address(0), "transfer to the zero address");

      uint256 senderBalance = balanceOf(sender);
      require(senderBalance >= amount, "transfer amount exceeds balance");
      unchecked {
         _subtractBalance(sender, amount);
      }
      _addBalance(recipient, amount);

      emit Transfer(sender, recipient, amount);
   }

   function totalSupply() public view override returns (uint256) {
      return _totalSupply;
   }

   function _mint(address account, uint256 amount) internal override {
      require(_totalSupply.add(amount) <= MAX_SUPPLY, "Max supply exceeded");
      _totalSupply += amount;
      _addBalance(account, amount);
      emit Transfer(address(0), account, amount);
   }

   receive() external payable {}
}
// File: @uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router02.sol



pragma solidity >=0.6.2;

import './IUniswapV2Router01.sol';

interface IUniswapV2Router02 is IUniswapV2Router01 {
    function removeLiquidityETHSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountETH);
    function removeLiquidityETHWithPermitSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountETH);

    function swapExactTokensForTokensSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
    function swapExactETHForTokensSupportingFeeOnTransferTokens(
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external payable;
    function swapExactTokensForETHSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
}

// File: @uniswap/v2-core/contracts/interfaces/IUniswapV2Pair.sol



pragma solidity >=0.5.0;

interface IUniswapV2Pair {
    event Approval(address indexed owner, address indexed spender, uint value);
    event Transfer(address indexed from, address indexed to, uint value);

    function name() external pure returns (string memory);
    function symbol() external pure returns (string memory);
    function decimals() external pure returns (uint8);
    function totalSupply() external view returns (uint);
    function balanceOf(address owner) external view returns (uint);
    function allowance(address owner, address spender) external view returns (uint);

    function approve(address spender, uint value) external returns (bool);
    function transfer(address to, uint value) external returns (bool);
    function transferFrom(address from, address to, uint value) external returns (bool);

    function DOMAIN_SEPARATOR() external view returns (bytes32);
    function PERMIT_TYPEHASH() external pure returns (bytes32);
    function nonces(address owner) external view returns (uint);

    function permit(address owner, address spender, uint value, uint deadline, uint8 v, bytes32 r, bytes32 s) external;

    event Mint(address indexed sender, uint amount0, uint amount1);
    event Burn(address indexed sender, uint amount0, uint amount1, address indexed to);
    event Swap(
        address indexed sender,
        uint amount0In,
        uint amount1In,
        uint amount0Out,
        uint amount1Out,
        address indexed to
    );
    event Sync(uint112 reserve0, uint112 reserve1);

    function MINIMUM_LIQUIDITY() external pure returns (uint);
    function factory() external view returns (address);
    function token0() external view returns (address);
    function token1() external view returns (address);
    function getReserves() external view returns (uint112 reserve0, uint112 reserve1, uint32 blockTimestampLast);
    function price0CumulativeLast() external view returns (uint);
    function price1CumulativeLast() external view returns (uint);
    function kLast() external view returns (uint);

    function mint(address to) external returns (uint liquidity);
    function burn(address to) external returns (uint amount0, uint amount1);
    function swap(uint amount0Out, uint amount1Out, address to, bytes calldata data) external;
    function skim(address to) external;
    function sync() external;

    function initialize(address, address) external;
}

// File: @uniswap/v2-core/contracts/interfaces/IUniswapV2Factory.sol



pragma solidity >=0.5.0;

interface IUniswapV2Factory {
    event PairCreated(address indexed token0, address indexed token1, address pair, uint);

    function feeTo() external view returns (address);
    function feeToSetter() external view returns (address);

    function getPair(address tokenA, address tokenB) external view returns (address pair);
    function allPairs(uint) external view returns (address pair);
    function allPairsLength() external view returns (uint);

    function createPair(address tokenA, address tokenB) external returns (address pair);

    function setFeeTo(address) external;
    function setFeeToSetter(address) external;
}

// File: @openzeppelin/contracts/utils/math/SafeMath.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (utils/math/SafeMath.sol)

pragma solidity ^0.8.0;

// CAUTION
// This version of SafeMath should only be used with Solidity 0.8 or later,
// because it relies on the compiler's built in overflow checks.

/**
 * @dev Wrappers over Solidity's arithmetic operations.
 *
 * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
 * now has built in overflow checking.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            uint256 c = a + b;
            if (c < a) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b > a) return (false, 0);
            return (true, a - b);
        }
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
     *
     * _Available since v3.4._
     */
    function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
            // benefit is lost if 'b' is also tested.
            // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
            if (a == 0) return (true, 0);
            uint256 c = a * b;
            if (c / a != b) return (false, 0);
            return (true, c);
        }
    }

    /**
     * @dev Returns the division of two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a / b);
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
     *
     * _Available since v3.4._
     */
    function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
        unchecked {
            if (b == 0) return (false, 0);
            return (true, a % b);
        }
    }

    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        return a + b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return a - b;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        return a * b;
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator.
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return a / b;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return a % b;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {trySub}.
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b <= a, errorMessage);
            return a - b;
        }
    }

    /**
     * @dev Returns the integer division of two unsigned integers, reverting with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a / b;
        }
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * reverting with custom message when dividing by zero.
     *
     * CAUTION: This function is deprecated because it requires allocating memory for the error
     * message unnecessarily. For custom revert reasons use {tryMod}.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(
        uint256 a,
        uint256 b,
        string memory errorMessage
    ) internal pure returns (uint256) {
        unchecked {
            require(b > 0, errorMessage);
            return a % b;
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
        }
        _balances[to] += amount;

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        _balances[account] += amount;
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
        }
        _totalSupply -= amount;

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @uniswap/v2-periphery/contracts/interfaces/IUniswapV2Router01.sol



pragma solidity >=0.6.2;

interface IUniswapV2Router01 {
    function factory() external pure returns (address);
    function WETH() external pure returns (address);

    function addLiquidity(
        address tokenA,
        address tokenB,
        uint amountADesired,
        uint amountBDesired,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB, uint liquidity);
    function addLiquidityETH(
        address token,
        uint amountTokenDesired,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external payable returns (uint amountToken, uint amountETH, uint liquidity);
    function removeLiquidity(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETH(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountToken, uint amountETH);
    function removeLiquidityWithPermit(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountA, uint amountB);
    function removeLiquidityETHWithPermit(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax, uint8 v, bytes32 r, bytes32 s
    ) external returns (uint amountToken, uint amountETH);
    function swapExactTokensForTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapTokensForExactTokens(
        uint amountOut,
        uint amountInMax,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);
    function swapExactETHForTokens(uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);
    function swapTokensForExactETH(uint amountOut, uint amountInMax, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapExactTokensForETH(uint amountIn, uint amountOutMin, address[] calldata path, address to, uint deadline)
        external
        returns (uint[] memory amounts);
    function swapETHForExactTokens(uint amountOut, address[] calldata path, address to, uint deadline)
        external
        payable
        returns (uint[] memory amounts);

    function quote(uint amountA, uint reserveA, uint reserveB) external pure returns (uint amountB);
    function getAmountOut(uint amountIn, uint reserveIn, uint reserveOut) external pure returns (uint amountOut);
    function getAmountIn(uint amountOut, uint reserveIn, uint reserveOut) external pure returns (uint amountIn);
    function getAmountsOut(uint amountIn, address[] calldata path) external view returns (uint[] memory amounts);
    function getAmountsIn(uint amountOut, address[] calldata path) external view returns (uint[] memory amounts);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}